

## Сумма цифр содержимого ячейки

Это глава из книги Билла Джелена [Гуру Excel расширяют горизонты: делайте невозможное с Microsoft Excel](#).

Задача: требуется построить формулу, которая возвращает сумму всех цифр в строке текста. Например, применяя формулу к текстовой строке «мне 24 года, а моему отцу – 43», хотелось бы получить 13 (2 + 4 + 4 + 3).

Решение. Вы знаете, что символы, которые участвуют в операции сложения, должны являться цифрами от 1 до 9. Поэтому алгоритм суммирования может быть следующим:

1 \* количество единиц в строке +

2 \* количество двоек в строке +

... +

9 \* количество девяток в строке = результат

Чтобы узнать количество вхождений цифры (скажем, 4), можно воспользоваться функцией ПОДСТАВИТЬ с заменой цифры 4 на пустую строку. Например, ПОДСТАВИТЬ(текст;4;"") вернет текст без всех четверок. В нашем примере – «мне 2 года, а моему отцу – 3». Если подсчитать разность длин исходной и получившейся строк, то это и будет количество четверок в строке: =ДЛСТР(A1)-ДЛСТР(ПОДСТАВИТЬ(A1;4;"")). Ответ = 2.

Чтобы обработать все цифры в текстовой строке, можно применить формулу массива: =ДЛСТР(A1)-ДЛСТР(ПОДСТАВИТЬ(A1;{1;2;3;4;5;6;7;8;9};"")). Этот фрагмент формулы вернет массив из девяти чисел. В нашем примере: {0;1;1;2;0;0;0;0;0}. Т.е., одна двойка, одна тройка и две четверки.

Таким образом, формула

=СУММ((ДЛСТР(A1)-ДЛСТР(ПОДСТАВИТЬ(A1;{1;2;3;4;5;6;7;8;9};""))) \* {1;2;3;4;5;6;7;8;9}) вернет сумму произведений двух массивов: =СУММ({0;1;1;2;0;0;0;0;0} \* {1;2;3;4;5;6;7;8;9}) =СУММ({0;2;3;8;0;0;0;0;0}) =13

Альтернативная стратегия: вместо функции СУММ можно использовать функцию СУММПРОИЗВ; при этом два массива следует разместить в двух аргументах функции: =СУММПРОИЗВ(ДЛСТР(A3)-ДЛСТР(ПОДСТАВИТЬ(A3;{1;2;3;4;5;6;7;8;9};""));{1;2;3;4;5;6;7;8;9}). В таком виде формула, вероятно, выглядит немного более понятной для большинства пользователей.

Изучите следующие примеры:

	A	B
1	мне 24 года, а моему отцу – 43	13
2		
3	76432	22
4	***** (8 121)	12
5	76*432	22
6	764 тест 32	22
7	1 тест 2	3
8	156 546	27
9	3127543,791	42
10	t=18317; p=239317	45
11	24 / 12	9
12	30°54'43"	19
13	SSN 421-89-7322	38
14	800/555-1212	29
15	3,1415926	77
16	18:00	12

Два последних примера, пожалуй, требуют пояснений. В ячейке A15 набрана формула =ПИ(). Сумма цифр в ячейке B15 – 77 – отражает сумму цифр числа  $\pi$  с максимально возможной точностью. Excel рассчитывает для числа  $\pi$  «лишь» 14 знаков после запятой. В ячейке A16 набрано время – 18:00. Если учесть, что Excel хранит дату и время, как порядковое число, то 18:00 соответствует  $\frac{3}{4}$  суток. В десятичной записи – это 0,75. Вот откуда взялся результат в B16 – 12 (7 + 5)

Резюме: вы можете использовать функции СУММ или СУММПРОИЗВ с элементами массивов, чтобы подсчитать сумму всех цифр в строке текста.

[Источник.](#)